

Un intelligente programmino per il Sinclair ZX-81 viene da Rijeka (Fiume), in Jugoslavia. L'autore, Petar Radovchich, ce lo invia senza alcun tipo di commento.

Dai dati numerici chiesti ed ottenuti nelle linee 1-4 il programma lista il mese desiderato, incolonnando per giorni della settimana. La linea 7 e la linea 13 mettono in luce un'istruzione nascosta dello ZX: il comando VAL "abcdefgh" (M) restituisce l'm-esimo numero contenuto nella stringa tra virgolette, ed è possibile anche richiedere più di uno dei caratteri della stringa, come mostrato in linea 13 (S-1 to S). Per pura curiosità abbiamo cercato il massimo numero di cifre che è possibile mettere nelle virgolette, ma oltre il 200° ci siamo stufati!

La linea 7, comunque, stabilisce — in base al numero d'ordine del mese "da 1 a 12) — la durata in giorni, sottraendo al numero 31 (massima durata di un mese) un appropriato valore. La linea 8 provvede ad incrementare di una unità la durata del febbraio bisestile: solo ogni quattro anni, infatti, $R / \text{INT}(R/4)$ è nullo.

Le linee 9 e 10 contengono delle correzioni di cui necessitano le successive tre linee per il calcolo del giorno di partenza (lunedì, martedì ...): il calcolo appare macchinoso, ma in un paio d'ore non ci è venuto in mente nulla di serio, tranne che dare al programma — come dato — che giorno sia il 1° gennaio dell'anno zero, e poi fargli calcolare il giorno iniziale del mese che ci interessa, ad es. con un LET GIORNIN=(anno *365 + INT(anno /4) - 1)/7, e poi confrontare per le coordinate appropriate.

Le coordinate sono date elementarmen- te nelle linee da 15 a 20: stabiliti i valori iniziali — calcolata quella della Y, data in linea 14 quella della X — basta organizzare un semplice 'a capo' ad ogni fine riga.

Con poche modifiche il programma può

essere adattato alla ZX PRINTER: con un ciclo si può ottenere la stampa di un intero calendario, e magari con qualche cambio nella formattazione della stampa può uscire qualche grafica simpatica. Le linee da 1

a 5 possono così essere eliminate e sostituite da un ciclo più un LET R=anno.

Ulteriori suggerimenti dei lettori saranno ovviamente i benvenuti.

Leo Sorge

```
ZX MINI CALENDAR ( ZX-81 1K RAM ) BY PETAR RADOVCHICH
```

```

1 PRINT AT 21,0;"MONTH ? ";
2 INPUT M
3 PRINT "YEAR ? ";
4 INPUT R
5 CLS
6 PRINT "SU MO TU WE TH FR SA -";M;"//";R
7 LET D=31-VAL "03101001010"(M)
8 IF M=2 AND R/4-INT(R/4)=0 THEN LET D=D+1
9 IF M<3 THEN LET R=R-1
10 IF M<3 THEN LET M=M+12
11 LET N=R+INT(R/4)+INT(R/400)-INT(R/100)+3*M+2-INT((2*M+1)/5)
12 LET S=((INT((N/7-INT(N/7))*10))+1)*2
13 LET Y=VAL "0369121518"(S-1 TO S)
14 LET X=2
15 FOR L=1 TO D
16 PRINT AT X,Y;L
17 LET Y=Y+3
18 IF Y>18 THEN LET X=X+2
19 IF Y>18 THEN LET Y=0
20 NEXT L
21 RUN

```

LA LINEA INCANCELLABILE

Se volete essere sicuri che il vostro programma porti il vostro nome, da chiunque venga copiato, oppure — più produttivamente — desiderate che i rapporti delle informazioni incancellabili per errore (nome, argomento, etc.) eccovi una semplice istruzione ad esecuzione immediata:

```
POKE 16510, 0
```

Modificando il contenuto della locazione 16510 si sposta la posizione della prima linea; così ponendo

```
1 REM "PROGRAMMA INCANCELLABILE"
```

```
    e poi POKE 16510,0, otterremo uno
```

```
0 REM "PROGRAMMA INCANCELLABILE",
```

linea inattaccabile da distrazioni. È da notare che non troveremo più la linea 1, e potremo riutilizzarla senza problemi.

Generalmente la locazione di memoria indicata contiene il valore 118, ma ponendo POKE 16510, 118 si porta il numero della prima linea a 118. Tra l'altro l'argomento della POKE accetta valori negativi, che vanno sottratti a 256: quindi

```
POKE 16510, -255
```

porta il numero della prima linea al valore $256-255=1$; però, il valore -256 dà una segnalazione di errore B.

L'autore della trovata è mister David Bayley di Leeds, ed è stata da noi vista su Practical Computing del Novembre 1981.

L.S.

INVIATECI I VOSTRI PROGRAMMI!

Se, qualunque sia la vostra macchina, avete realizzato programmi o routine che ritenete possano interessare altri lettori, inviateceli. Saranno esaminati e, se pubblicati, ricompensati con valutazioni approssimativamente fra le 30 e le 100.000 lire, secondo la complessità, la genialità, l'originalità e la presentazione del materiale e della documentazione (listati, diagrammi, commenti ecc.). Per ragioni organizzative non possiamo impegnarci, salvo eventuali accordi presi prima dell'invio, alla restituzione dei materiali, che resteranno di proprietà della redazione che si impegna a non divulgarli (se non tramite la rivista) senza l'autorizzazione dei rispettivi autori.